

METODO PARA EL CALCULO EN "ORDENADORES PERSONALES" DE LOS "VALORES ESPERADOS" DE VARIABLES ANTROPOMETRICAS SOBRE COMPOSICION CORPORAL

Isabel Goñi¹ y Luis García-Diz²

**Instituto de Nutrición (C. S. I. C.), Facultad de Farmacia,
Universidad Complutense, Madrid, España**

RESUMEN

Se presenta un programa para incorporar a cualquier ordenador personal (micro-procesador) que posea lenguaje BASIC. Dicho programa permite calcular los valores esperados (percentil 50) y el rango de variación "admisible" (percentiles 3 y 97) de los espesores de los pliegues cutáneos bicipital, tricipital y subescapular, así como del perímetro del brazo, tanto en niños como en niñas con edades comprendidas entre los 6 y 14 años. Es fácilmente modificable a cualquier población estudiada, y elabora también un diagnóstico antropométrico.

Con este programa se facilita enormemente el manejo e interpretación de algunas variables biométricas usuales, relacionadas con la composición corporal, necesarias para el análisis completo del estado nutricional actual, tanto en individuos como en colectividades.

INTRODUCCION

Diagnosticar el estado nutricional de un grupo, e incluso el de cada individuo, a partir de variables biométricas es una de las principales posibilidades de la antropometría (1-4). Ello es posible gracias al uso de material de precisión especialmente diseñado para este fin, así como al control riguroso del personal que interviene en la recolección directa de los datos de campo.

Por razones obvias, las variables más utilizadas para estos fines han sido, son y serán, el peso y la talla. Pero junto a estos parámetros clásicos se han de añadir, hoy en día, la determinación del espesor de los pliegues cutáneos como medida directa de la grasa periférica (2, 5, 6) y algunos perímetros corporales indicativos del estado general del músculo esquelé-

Manuscrito modificado recibido: 15-2-85.

1 Profesor Ayudante de Fisiología y Nutrición de la Facultad de Farmacia, Universidad Complutense, Madrid 28040, España.

2 Profesor Titular de Nutrición y Bromatología de la Facultad de Farmacia, Universidad Complutense de Madrid.

tico (2, 6-8). Este tipo de variables antropométricas también necesitan ser contrastadas con datos de "referencia" correspondientes a una población "bien nutrida" que señale los valores que cabría esperar para cada edad, sexo, grupo étnico, etc. Este aspecto se halla parcialmente cubierto por la bibliografía específica existente (2, 5-10), aunque todavía quedan lagunas que cubrir.

Sea como fuere, el manejo de tablas de referencia con una gran cantidad de datos, se hace engorroso en la mayoría de las ocasiones. Dificulta igualmente, su incorporación a los "ordenadores personales", cuyo uso se está imponiendo en muchos laboratorios de nutrición (11, 12).

La presentación, junto con las tablas de valores de referencia, de una función sencilla que sea fácilmente programable y que permita el cálculo del valor esperado para cada variable antropométrica, enriquecería y facilitaría, en gran manera, las posibilidades de trabajo de antropometristas y nutriólogos. De la misma forma, ayudaría a la realización de diagnósticos nutricionales inmediatos, por comparación entre los valores que se posean como reales, y los calculados como "esperados" para ese caso. El uso de estas ecuaciones permitiría también realizar interpolaciones precisas para cualquier edad concreta, en años, meses y días. Se evitaría, así, las aproximaciones a un año, o los cálculos de incrementos proporcionales a la edad en variables que no presentan evoluciones tan sencillas como es el caso del espesor del panículo adiposo (5), y que suelen hacerse al manejar las tablas de referencia.

Por todo ello hemos creído de interés continuar suministrando (4), junto a los clásicos valores experimentales obtenidos, una serie de "algoritmos" que permitan el cálculo de los valores "esperados" de algunas variables antropométricas, correspondientes a escolares madrileños de ambos sexos y edades comprendidas entre los 6 y 14 años. Se incluye además, rutinas realizadas en BASIC en las que están programados dichos algoritmos, así como las indicaciones necesarias para su adaptación a otras poblaciones estudiadas. El fin que se persigue es que puedan ser incorporadas a cualquier microprocesador que disponga de este lenguaje, ya sea tan solo para el cálculo de los valores esperados para cada edad y sexo, o para ayudar a una evaluación antropométrica del estado nutricional de individuos o colectividades.

MATERIAL Y METODOS

El material que se presenta en este trabajo forma parte de una investigación más amplia sobre la alimentación de los escolares españoles, y algunas de sus consecuencias. El estudio lo está llevando a cabo actualmente nuestro laboratorio, y sus resultados se han comenzado a presentar en diversas publicaciones (4, 13-15).

Los sujetos estudiados tenían edades comprendidas entre los 6 y 14 años y cursaban sus estudios de educación general básica (E.G.B.), en un colegio situado en el casco urbano de Madrid. Se seleccionaron mediante sorteo 400 niños y niñas, estableciéndose 16 grupos provisionales de 25 personas de cada sexo por cada uno de los ocho niveles de que consta la E.G.B. La falta de aprobación requerida, por parte de padres o tutores, en cuanto a la participación de sus hijos en el estudio, o la inasistencia de

algunos sujetos al colegio los días en que se realizó la toma de datos, redujo el número inicial de seleccionados. En resumen, pudimos contar con 164 niños y 169 niñas con los que establecimos los grupos definitivos por edades y sexo.

La toma de datos se efectuó en las propias instalaciones del colegio, durante los meses de febrero y marzo de 1983. Para este propósito se desplazó el personal y material necesarios para realizar las medidas biométricas, que se efectuaron por las mañanas en los sujetos, en ayunas y desnudos.

Los parámetros obtenidos en cada uno, y que son los utilizados en este informe, fueron: espesores de los pliegues cutáneos bicipital (EsB), tricipital (EsT) y subescapular (EsS) y circunferencia del brazo (CB). Todos ellos fueron tomados en el lado izquierdo del cuerpo, y en los lugares apropiados (1), en duplicado y por dos personas entrenadas según los criterios internacionales (16, 17). Para su realización se contó con un lipocalibre de presión constante Holtain (precisión 0.2 mm) y una cinta métrica Medicon Instruments (precisión 1 mm).

Los datos globales se agruparon por sexo y edades, presentándose en las Tablas 1 y 2 los valores correspondientes a los percentiles 3, 50 y 97, obtenidos de los datos originales (14) por los procedimientos usuales (18). En el caso concreto de los espesores cutáneos, y puesto que presentan una distribución logarítmica normal (2, 5), se efectuó una transformación logarítmica previa. Se calcularon entonces los percentiles indicados, deshaciéndose la transformación nuevamente para presentar los resultados finales.

Los cálculos de los algoritmos de trabajo (polinómicas, semilogarítmicas, logarítmicas y exponenciales) entre las distintas variables, se hicieron por el método de máxima verosimilitud (18, 19). Se contó con el auxilio de un microprocesador CBM 4032 y programas de ajustes multivariantes (20, 21) usando como criterio de selección, entre las distintas funciones halladas, los valores mínimos del error estándar (ES) y máximos del coeficiente de determinación (R^2) de cada una. Se siguió la rutina de trabajo y criterios más usuales para el ajuste de funciones (18, 19).

Este mismo microprocesador fue el utilizado para realizar y comprobar el correcto funcionamiento del programa de estimación de los valores de referencia y el diagnóstico antropométrico que se ofrece al final de este artículo.

RESULTADOS Y DISCUSION

Los valores correspondientes a todas y cada una de las variables antropométricas obtenidas en la muestra de escolares madrileños se resumen en las Tablas 1 y 2. Estos datos no difieren significativamente de los señalados por otros autores para poblaciones infantiles europeas similares a la española (9, 13, 14). A ello se une un estado nutricional adecuado, juzgado por las técnicas de encuestas de alimentación (15). Por estos motivos, decidimos utilizarlos como valores de partida adecuados para mostrar la metodología de trabajo a seguir con miras de informatizar los datos antropométricos de este grupo de escolares madrileños, base del cálculo de los valores esperados, y diagnóstico nutricional posterior.

TABLA 1

VALORES DEL ESPESOR DEL PANICULO ADIPOSO Y PERIMETRO BRAQUIAL EN NIÑOS DE DISTINTAS EDADES

Edad años	No.	EsB (mm)			EsT (mm)			EsS (mm)			CB (cm)		
		p 3	p 50	p 97	p 3	p 50	p 97	p 3	p 50	p 97	p 3	p 50	p 97
6- 7	18	3.84	6.59	11.30	4.37	7.85	14.10	3.35	5.49	8.99	14.59	17.61	20.63
7- 8	21	3.75	7.56	15.24	4.19	8.34	16.58	3.02	5.93	11.62	15.05	19.11	23.17
8- 9	16	3.14	7.75	19.10	3.39	8.67	22.19	2.54	6.26	15.43	13.36	19.62	25.88
9-10	25	3.38	7.82	18.07	3.67	8.36	19.06	3.23	6.60	13.50	14.90	20.16	25.42
10-11	17	2.21	5.58	14.10	3.78	8.28	18.14	2.91	6.81	15.94	16.42	21.38	26.34
11-12	25	2.53	5.77	13.16	4.27	9.22	19.93	3.04	7.89	20.45	16.94	22.06	27.18
12-13	23	1.94	5.53	15.79	3.22	8.88	24.46	2.66	7.96	23.82	16.51	22.53	28.55
13-14	19	1.90	4.97	13.04	3.34	7.72	17.84	3.27	7.26	16.12	17.24	22.96	28.68

Los valores representan los percentiles 3, 50 y 97, respectivamente, de los espesores de los pliegues cutáneos bicipital (EsB), tricípital (EsT) y subescapular (EsS), así como de la circunferencia del brazo (CB).

Datos presentados para su publicación en la Revista Clínica Española, 1984 (14).

TABLA 2

VALORES DEL ESPESOR DEL PANICULO ADIPOSEO Y PERIMETRO BRAQUIAL EN NIÑAS DE DISTINTAS EDADES

Edad años	No.	EsB (mm)			EsT (mm)			EsS (mm)			CB (cm)		
		p 3	p 50	p 97	p 3	p 50	p 97	p 3	p 50	p 97	p 3	p 50	p 97
6-7	19	4.61	8.03	13.99	5.17	9.28	16.66	3.63	6.24	10.71	14.26	18.20	22.14
7-8	20	4.61	7.79	13.16	6.11	10.17	16.92	3.81	6.74	11.92	15.44	18.78	22.12
8-9	17	4.43	7.49	12.66	6.11	10.17	16.92	3.81	6.74	11.92	15.33	19.65	23.97
9-10	22	3.87	7.49	14.47	5.66	10.33	18.82	3.57	7.30	14.93	15.83	20.23	24.63
10-11	21	3.31	6.76	13.82	5.46	10.70	20.97	3.57	7.30	14.93	17.44	21.52	25.60
11-12	18	3.17	6.75	14.38	5.51	11.42	23.68	3.40	9.71	27.73	17.20	22.32	27.44
12-13	22	3.11	6.28	12.66	5.26	10.46	20.80	4.07	8.68	18.50	16.96	22.42	27.88
13-14	30	2.85	5.28	9.77	5.51	9.60	16.73	4.98	9.08	16.55	19.08	22.98	26.88

Los valores representan los percentiles 3, 50 y 97, respectivamente, de los espesores de los pliegues cutáneos bicipital (EsB), tricipital (EsT) y subescapular (EsS), así como de la circunferencia del brazo (CB).

Datos presentados para su publicación en la Revista Clínica Española, 1984 (14).

A partir de estas tablas iniciales (Tablas 1 y 2), nos propusimos encontrar una serie de funciones matemáticas, lo más sencillas posible, que permitieran reproducir los valores experimentales hallados para la población estudiada.

Realizamos así un programa de ordenador que calculara, por el método de máxima verosimilitud, toda una serie de funciones lineales, polinómicas, semilogarítmicas, exponenciales y doble logarítmicas para ajustar los valores experimentales de cada una de las variables antropométricas a la edad de cada niño en función de su sexo (20, 21). Todas las funciones descritas fueron halladas, escogiéndose, en cada caso, aquella que presentaba el máximo coeficiente de determinación (fracción de la varianza de la variable dependiente explicada por la función elegida), junto con el menor error estándar posible (18, 19).

El grupo de funciones que mejor se adaptó a los valores experimentales fueron, en todos los casos estudiados, las polinómicas, presentándose todas en las Tablas 3, 4 y 5. En ellas figuran los coeficientes de las funciones polinómicas de grado 5 ó menos, correspondientes a las curvas de evolución con la edad, de los valores medianos (percentil 50) de cada variable antropométrica, para cada uno de los sexos independientemente, junto con los correspondientes a los percentiles 3 y 97 de las mismas. A partir de estas ecuaciones es factible calcular el valor esperado (percentil 50) y el rango "admisible" (percentiles 3 y 97) de cada variable, para una edad y sexo determinados. Además, todos los valores contenidos en las tablas de partida (Tablas 1 y 2) pueden reproducirse.

El siguiente paso consistió en realizar un pequeño programa, para un ordenador personal, que permitiera automatizar el análisis y diagnóstico antropométrico de los grupos estudiados.

El programa que se da a conocer aquí ha sido realizado en BASIC y diseñado en forma que pueda ser usado para poblaciones estudiadas distintas de la nuestra. Lo único que se necesita es ajustar los datos de la nueva población deseada a una función polinómica (16-19) e introducir los grados de dichos polinomios y los valores de los coeficientes hallados en las sentencias DATA comprendidas entre las líneas de programa 1100 a 1300.

Para efectuar dicha corrección deben ajustarse los valores correspondientes a los percentiles 97, 50 y 3 de la población a estudiar, de los espesores de los pliegues cutáneos bicipital, tricípital y subescapular de los niños y después los de las niñas. Es imprescindible realizar estas operaciones en el orden indicado, con el fin de construir nuevas tablas de coeficientes equivalentes a las Tablas 3 y 4 que aquí se incluyen. De todas las funciones encontradas, se escoge como grado de ajuste (GP) el mayor de todos, completándose con "0" los coeficientes que falten en las funciones de grado inferior, tal como se observa en el percentil 3 del espesor bicipital de niños (Tabla 3). Una vez hecho esto, se deben sustituir las sentencias DATA del programa comprendidas entre la 1100 y 1200, introduciéndose el grado máximo (GP) en la primera de ellas (1100), y a continuación, y en el orden ya mencionado, los coeficientes de los polinomios de ajuste, desde el término independiente hasta el de mayor grado en orden creciente, sin olvidar los ceros.

Luego, los datos correspondientes al contorno de brazo (percentiles 97, 50 y 3) para niños y niñas, deben ajustarse igualmente, siguiéndose los

TABLA 3
COEFFICIENTES DE LAS FUNCIONES POLINOMICAS PARA CALCULAR LOS VALORES ESPERADOS DE LOS
ESPORES DE LOS PLIEGUES CUTANEOS EN NIÑOS DE 6 A 14 AÑOS

	A grado 0	B grado 1	C grado 2	D grado 3	E grado 4	F grado 5	R ²	ES
EsB								
percentil 97	3024.46	-1689.39	369.35	-39.3683	2.0499	-0.0418	0.957	1.025
percentil 50	581.33	-326.75	72.26	-7.7672	0.4076	-0.0083	0.913	0.629
percentil 3	-1.12	1.99	-0.24	0.0083	0.0000	0.0000	0.899	0.334
EsT								
percentil 97	3725.18	-2090.71	461.00	-49.7439	2.6304	-0.0546	0.920	1.705
percentil 50	-2.04	-9.62	5.24	-0.8779	0.0621	-0.0016	0.868	0.340
percentil 3	-647.26	355.40	-75.88	7.9298	-40.61	0.0082	0.840	0.339
EsS								
percentil 97	1535.76	-916.03	214.06	-24.3905	1.3586	-0.0296	0.977	1.346
percentil 50	3.40	-7.56	3.49	-0.5698	0.0405	-0.0011	0.980	0.234
percentil 3	-165.72	101.65	-23.67	2.6726	-0.1468	0.0031	0.665	0.313

Función = $A + B Ed + C Ed^2 + D Ed^3 + E Ed^4 + F Ed^5$ (siendo Ed = edad del sujeto en años (fracción decimal), R² = coeficiente de determinación, y ES = error estándar).

TABLA 4

COEFFICIENTES DE LAS FUNCIONES POLINOMICAS PARA CALCULAR LOS VALORES ESPERADOS DE LOS
ESPEORES DE LOS PLIEGUES CUTANEOS EN NIÑAS DE 6 A 14 AÑOS

	A grado 0	B grado 1	C grado 2	D grado 3	E grado 4	F grado 5	R ²	ES
EsB								
percentil 97	1446.90	-777.55	165.43	-17.2669	0.8856	-0.0179	0.816	1.604
percentil 50	159.24	-82.88	17.96	-1.9221	0.1013	-0.0021	0.983	0.221
percentil 3	-239.93	130.71	-27.24	2.7725	-0.1383	0.0027	0.853	0.489
EsT								
percentil 97	-105.64	59.48	-10.82	0.8656	-0.0253	0.0000	0.913	1.125
percentil 50	-541.54	286.41	-58.48	5.8618	-0.2881	0.0056	0.937	0.306
percentil 3	-486.27	254.85	-52.01	5.2312	-0.2596	0.0051	0.721	0.283
EsS								
percentil 97	-6688.91	3675.93	-791.41	83.4902	-4.3126	0.0873	0.916	3.752
percentil 50	-1547.93	846.86	-181.12	18.9933	-0.9761	0.0197	0.940	0.737
percentil 3	-236.60	126.83	-26.27	2.6742	-0.1341	0.0026	0.948	0.223

Función = $A + B Ed + C Ed^2 + D Ed^3 + E Ed^4 + F Ed^5$ (siendo Ed = edad del sujeto en años (fracción decimal), R² = coeficiente de determinación, y ES = error estándar).

TABLA 5

COEFICIENTES DE LAS FUNCIONES POLINOMICAS PARA CALCULAR LOS VALORES ESPERADOS DE LA CIRCUNFERENCIA DEL BRAZO EN NIÑOS Y NIÑAS DE 6 A 14 AÑOS

	A grado 0	B grado 1	C grado 2	D grado 3	E grado 4	F grado 5	R ²	ES
CB en niños								
percentil 97	-167.19	72.34	-10.25	0.6460	-0.0151	0.0000	0.977	0.642
percentil 50	-448.39	239.39	-48.52	4.8509	-0.2385	0.0046	0.997	0.186
percentil 3	-1266.09	694.03	-147.35	15.3204	-0.7805	0.0156	0.931	0.651
CB en niñas								
percentil 97	61.74	-14.42	1.66	-0.0576	0.0000	0.0000	0.976	0.464
percentil 50	27.30	-4.01	0.52	-0.0184	0.0000	0.0000	0.991	0.223
percentil 3	-1084.31	596.07	-127.43	13.4157	-0.6951	0.0142	0.974	0.456

Función = $A + B Ed + C Ed^2 + D Ed^3 + E Ed^4 + F Ed^5$ (siendo Ed = edad del sujeto en años (fracción decimal), R² = coeficiente de determinación, y ES = error estándar).

critérios ya mencionados para el caso anterior, lo que llevaría a confeccionar una nueva tabla, análoga a la Tabla 5. El máximo grado de las seis funciones calculadas (GB), se introducirá en la primera sentencia DATA de la sección comprendida entre las líneas de programa 1200 a 1300, y a continuación, los valores de los coeficientes hallados, cuidando el orden.

El programa así modificado, es muy útil. Puede emplearse para la estimación de los valores medianos y rango válido "admisible" de cualquier grupo poblacional, realizando la comparación entre el valor observado en cada individuo y el que "debería" tener, como base del diagnóstico antropométrico.

```

1 REM *****
2 REM *          PROGRAMA DE CONTROL PRINCIPAL          *
3 REM *****
4 REM
9 REM          *****
10 DIM P(18), C(16) : REM *      VECTORES DE LOS VALORES DE      *
11 REM *          REFERENCIA CALCULADOS DE              *
12 REM *          LOS PLIEGUES CUTANEOS (P), Y          *
13 REM *          CONTORNO DEL BRAZO (C)                *
14 REM          *****
16 DEFFND(A) = INT(A*100 +.5)/100 : REM *      AJUSTA LOS DATOS      *
18 REM *          A 2 DECIMALES                          *
19 REM          *****
20 GOSUB 1000 : REM *      CARGA LA MATRIZ DE COEFICIENTES      *
29 REM          *****
30 GOSUB 1300 : REM *      PIDE LOS DATOS CONCRETOS DE CADA      *
31 REM *          SUJETO EN PARTICULAR                  *
39 REM          *****
40 GOSUB 1500 : REM *      CALCULA LOS VALORES ESPERADOS        *
49 REM          *****
50 GOSUB 1600 : REM *      COMPARA LOS DATOS OBTENIDOS CON      *
51 REM *          LOS ESPERADOS, Y REALIZA UN            *
52 REM *          DIAGNOSTICO                            *
59 REM          *****
60 GOSUB 2000 : REM *      MUESTRA LOS RESULTADOS FINALES      *
68 REM          *****
69 REM
70 PRINT "DESEA CONTINUAR CON OTRO SUJETO? SI O NO"
72 GET A$: IF A$ <> "" THEN 72
74 GET A$: IF A$ = "" THEN 74
76 IF A$ = "S" THEN 30
78 IF A$ = "N" THEN END
80 GOTO 74

988 REM *****
989 REM *          RUTINA PARA CARGAR LOS COEFICIENTES          *
990 REM *          DE LAS ECUACIONES DE AJUSTE                *
991 REM *          GP = GRADO MAYOR DE LAS FUNCIONES DE AJUSTE *
992 REM *          DE LOS PLIEGUES CUTANEOS                  *
993 REM *          CP (18, GP +1) = MATRIZ DE COEFICIENTES DE  *

```

```

994 REM *           LOS PLIEGUES CUTANEOS *
995 REM *           GB = GRADO MAYOR DE LAS FUNCIONES DE AJUSTE *
996 REM *           DEL CONTORNO DEL BRAZO *
997 REM *           CB (6, GB + 1) = MATRIZ DE COEFICIENTES *
998 REM *           DEL CONTORNO DEL BRAZO *
999 REM *****
1000 READ GP
1010 DIM CP (18, GP + 1)
1020 FOR I = 0 TO 17
1025 FOR J = 0 TO GP
1030 READ CP (I, J)
1035 NEXT J
1040 NEXT I
1050 READ GB
1060 DIM CB (6, GB + 1)
1070 FOR I = 0 TO 1
1075 FOR J = 0 TO GB
1080 READ CB (I, J)
1085 NEXT J
1090 NEXT I
1094 REM *****
1095 REM * SENTENCIAS DONDE SE ALMACENAN LOS VALORES DEL *
1096 REM * GRADO Y COEFICIENTES DE LAS FUNCIONES DE *
1097 REM * AJUSTE DE LA POBLACION DE REFERENCIA *
1098 REM *****
1099 REM *****VALORES DE AJUSTE DE LOS PLIEGUES CUTANEOS*****
1100 DATA 5
1109 REM *****DATOS DE PLIEGUES CUTANEOS EN NIÑOS*****
1110 DATA 3024.46,-169.39,369.35,-39.3683,2.0499,-.0418
1112 DATA 581.33,-326.75,72.26,-7.7672,.4076,-.0083
1114 DATA -1.12,1.99,-.24,.0083,0,0
1116 DATA 3725.18,-2090.71,461,-49.7439,2.6304,-.0546
1118 DATA -2.04,-9.62,5.24,-.8779,.0621,-.0016
1120 DATA -647.26,355.4,-75.88,7.9298,-.4061,.0082
1122 DATA 1535.76,-916.03,214.06,-24.3905,1.3586,-.0296
1124 DATA 3.40,-7.56,3.49,-.5698,.0405,-.0011
1126 DATA -165.72,101.62,-23.67,2.6726,.1468,.0031
1127 REM *****DATOS DE PLIEGUES CUTANEOS EN NIÑAS*****
1128 DATA 1446.9,-777.55,165.43,-17.2669,.8856,-.0179
1130 DATA 159.24,-82.88,17.96,-1.9221,0.1013,-.0021
1132 DATA -239.93,130.71,27.24,2.7725,-.1383,0.027
1134 DATA -105.64,59.48,-10.82,.8656,-.0256,0
1136 DATA -541.54,286.41,-58.48,5.8618,-.2881,.0056
1138 DATA -486.27,254.85,-52.01,5.2312,-.2596,.0051
1140 DATA -6688.91,3675.93,-791.41,83.4902,-4.3126,.0873
1142 DATA -1547.93,846.86,-181.12,18.9933,-.9761,.0197
1144 DATA -236.6,126.83,-26.27,2.6742,-.1341,.0026
1149 REM *****VALORES DE AJUSTE DEL CONTORNO DE BRAZO*****
1150 DATA 5
1152 REM *****DATOS CORRESPONDIENTES A VARONES*****
1160 DATA -167.19,72.34,-10.25,.646,-.0151,0

```

```

1165 DATA -448.39,239.39,-48.52,4.8509,-.2385,.0046
1170 DATA -1266.09,694.03,-147.35,15.3204,-.7805,.0156
1172 REM *****DATOS CORRESPONDIENTES A MUJERES*****
1175 DATA 61.74,-14.42,1.66,-.0576,0,0
1180 DATA 27.30,-4.01,.52,-.0184,0,0
1185 DATA -1084.31,596.07,-127.43,13.4157,-.6951,.0142
1287 REM
1288 REM *****
1289 REM *   RUTINA PARA INTRODUCIR EN EL ORDENADOR LOS   *
1290 REM *   DATOS OBTENIDOS DE UN SUJETO CONCRETO       *
1291 REM *   ED = EDAD EN AÑOS (FRACCION DECIMAL)*      *
1292 REM *   SX$ = SEXO (VARON O MUJER)                  *
1293 REM *   PB = ESPESOR DEL PLIEGUE BICIPITAL          *
1294 REM *   PT = ESPESOR DEL PLIEGUE TRICIPITAL        *
1295 REM *   PS = ESPESOR DEL PLIEGUE SUBESCAPULAR      *
1296 REM *   CB = CONTORNO DEL BRAZO                    *
1297 REM * SE EFECTUAN DIVERSAS COMPROBACIONES PARA EVITAR *
1298 REM * EN LO POSIBLE, LA INTRODUCCION DE DATOS ERRONEOS *
1299 REM *****
1300 INPUT "EDAD (EN AÑOS Y FRACCION DECIMAL) : " : ED
1310 IF (ED>6) AND (ED<14) THEN 1350
1320 PRINT "RANGO DE EDADES PERMITIDO ENTRE 6 Y 14"
1330 PRINT "VUELVA A INTRODUCIR LA EDAD"
1340 GOTO 1300
1350 INPUT "SEXO (VARON O MUJER) : " : SX$
1360 IF SX$ = "VARON" THEN II = 0 : GOTO 1400
1365 IF SX$ = "MUJER" THEN II = 1 : GOTO 1400
1370 PRINT "DENOMINACION DE LOS SEXOS : VARON O MUJER"
1380 PRINT "VUELVA A INTRODUCIR EL SEXO"
1390 GOTO 1350
1400 INPUT "PLIEGUE CUTANEO BICIPITAL (MM) : " : PB
1405 IF (PB>0) AND (PB<40) THEN 1425
1410 PRINT "RANGO DE MEDIDA DEL LIPOCALIBRE 0-40"
1415 PRINT "VUELVA A INTRODUCIR EL VALOR"
1420 GOTO 1400
1425 INPUT "PLIEGUE CUTANEO TRICIPITAL (MM) : " : PT
1430 IF (PT>0) AND (PT<40) THEN 1450
1435 PRINT "RANGO DE MEDIDA DEL LIPOCALIBRE 0-40"
1440 PRINT "VUELVA A INTRODUCIR EL VALOR"
1445 GOTO 1425
1450 INPUT "PLIEGUE CUTANEO SUBESCAPULAR (MM) : " : PS
1455 IF (PS>0) AND (PS<40) THEN 1475
1460 PRINT "RANGO DE MEDIDA DEL LIPOCALIBRE 0-40"
1465 PRINT "VUELVA A INTRODUCIR EL VALOR"
1470 GOTO 1450
1475 INPUT "CONTORNO DE BRAZO (CM) : " : CB
1480 IF CB>0 THEN RETURN
1485 PRINT "NO PUEDEN EXISTIR VALORES NEGATIVOS"
1490 PRINT "VUELVA A INTRODUCIR EL VALOR"
1492 GOTO 1475

```

```

1494 REM *****
1495 REM * RUTINA DE CALCULO DE LOS VALORES ESPERADOS PARA *
1496 REM * LA EDAD Y SEXO DEL SUJETO ESTUDIADO. LOS DATOS *
1497 REM * HALLADOS SE GUARDAN EN LOS VECTORES P(I) Y C(I) *
1498 REM * ** = CLAVE DE EXPONENCIACION *
1499 REM *****
1500 FOR I =  $\Pi * 9$  TO  $\Pi * 9 + 8$ 
1505 P(I) = 0
1510 FOR J = 0 TO GP
1520 P(I) = P(I) + CP(I,J) * ED ** J
1530 NEXT J
1540 NEXT I
1550 FOR I =  $\Pi * 3$  TO  $\Pi * 3 + 2$ 
1555 C(I) = 0
1560 FOR J = 0 TO GB
1570 C(I) = C(I) + CB(I,J) * ED ** J
1580 NEXT J
1590 NEXT I
1592 RETURN
1595 REM *****
1596 REM * RUTINA DE DIAGNOSTICO ANTROPOMETRICO *
1597 REM * COMPARACION ENTRE LOS VALORES REALES Y LOS *
1598 REM * CALCULADOS COMO ESPERADOS PARA EL SUJETO *
1599 REM *****
1600 D1$ = "ADMISIBLE"
1604 D2$ = "EXCESIVO"
1608 D3$ = "DEFICIENTE"
1610 DB$ = D1$
1614 IF PB > = P(0 +  $\Pi * 9$ ) THEN DB$ = D2$
1618 IF PB < = P(2 +  $\Pi * 9$ ) THEN DB$ = D3$
1620 DT$ = D1$
1624 IF PT > = P(3 +  $\Pi * 9$ ) THEN DT$ = D2$
1628 IF PT < = P(5 +  $\Pi * 9$ ) THEN DT$ = D3$
1630 DS$ = D1$
1634 IF PS > = P(6 +  $\Pi * 9$ ) THEN DS$ = D2$
1638 IF PS < = P(8 +  $\Pi * 9$ ) THEN DS$ = D3$
1640 DC$ = D1$
1644 IF CB > = C(0 +  $\Pi * 3$ ) THEN DC$ = D2$
1648 IF CB < = C(2 +  $\Pi * 3$ ) THEN DC$ = D3$
1650 RETURN
1991 REM *****
1995 REM * RUTINA DE SALIDA DE LOS RESULTADOS *
1996 REM * PREVISTA PARA MOSTRAR LOS RESULTADOS EN: *
1997 REM * LA PANTALLA DEL MICROORDENADOR --> 1 *
1998 REM * POR UNA IMPRESORA ACOPLADA --> 2 *
1999 REM *****
2000 PRINT "DESEA QUE LOS RESULTADOS SEAN MOSTRADOS POR :?"
2004 PRINT " 1 LA PANTALLA"
2008 PRINT " 2 LA IMPRESORA"
2010 GET A$ : IF A$ <> " " THEN 2010
2011 GET A$ : IF A$ = " " THEN 2011

```

```

2015 IF A$ = "1" THEN AR = 3 : GOTO 2020
2016 IF A$ = "2" THEN AR = 4 : GOTO 2020
2019 GOTO 2011
2020 OPEN 4, AR
2030 PRINTL4,TAB (10);"V.RE.";TAB(15);"V.ES.";TAB(20);
2031 PRINTL4,TAB(20);"RANGO FIS.";TAB(30);"DIAGNOS.—"
2032 PRINTL4,TAB(10);"-----"
2040 PRINTL4,"PL. BICIP."; FND(PB);TAB(15);FND(P(1+II*9));
2044 PRINTL4,TAB(20);FND(P(0+II*9));TAB(25);FND(P(2+II*9));
2048 PRINTL4,TAB(30);DB$
2050 PRINTL4,"PL. TRICP.";FND(PT);TAB(15);FND(P(4+II*9));
2054 PRINTL4,TAB(20);FND(P(3+II*9));TAB(25);FND(P(5+II*9));
2058 PRINTL4,TAB(30);DT$
2060 PRINTL4,"CIR. BRAZO";FND(CB);TAB(15);FND(C(1+II*3));
2064 PRINTL4,TAB(20);FND(C(0+II*));TAB(25);FND(C(2+II*3));
2068 PRINTL4,TAB(30);DC$
2070 CLOSE AR
2090 RETURN

```

SUMMARY

METHOD FOR CALCULATION, WITH "PERSONAL COMPUTERS" OF "EXPECTED VALUES" OF ANTHROPOMETRIC VARIABLES ON BODY COMPOSITION

A program to incorporate in any personal computer (microprocessor) with BASIC language, is herein proposed. This enables calculation of the expected values (50 percentile), and the range of "allowable" variables (3 and 97 percentiles) of the bicipital, tricipital and subscapular cutaneous skinfold thickness, and of the arm's circumference in 6 to 14-year-old boys and girls. It is easily modifiable so as to apply it to other population sectors, and also makes an anthropometric diagnosis.

The program facilitates the handling and interpretation of some current biometric variables related to body composition, which are necessary for the full analysis of the actual nutritional status, both of individuals and of population groups.

BIBLIOGRAFIA

1. Weiner, J. S. & C. H. Lourie. *Human Biology: A Guide to Field Methods*. Oxford, Blackwell Scientific Publications, 1969. (IBP Handbook No. 98).
2. Jelliffe, D. B. & E. F. P. Jelliffe. *Human Nutrition. Nutrition and Growth*. Vol. 2. New York and London, Plenum Press, 1980.
3. Tojo, R. Valoración del estado nutritivo. *Nutrición Clínica*, 3: 26-46, 1983.
4. García-Diz, L., I. Goñi & G. Varela. Aplicación del cálculo de valores antropométricos mediante microprocesador al diagnóstico nutricional. *Arch. Latinoamer. Nutr.*, 35: 48-62, 1985.
5. Roche, A. F., R. M. Siervogel, W. C. Chumlea, R. B. Reed, D. Eichorn & R. M. Mc Cammon. *Serial Changes in Subcutaneous Fat Thickness of Children and Adults*. Ed. Falkner, Dretchmer & Rossi, 1982. (Monographs in Pediatrics).

6. Ramos Galván, R. Somatometría pediátrica. Estudio semi-longitudinal en niños de la ciudad de México. *Archivos de Investigación Médica*, 6: Supl. 11, 1975.
7. Chávez, A. & C. Martínez. *Nutrición y Desarrollo Infantil*. México D. F., México, Nueva Editorial Interamericana, 1979.
8. Jordan, J. R. *Desarrollo Humano en Cuba*. La Habana, Editorial Científico Técnica, 1979.
9. Eveleth, P. B. E. & J. M. Tanner. *Worldwide Variation in Human Growth*. Cambridge, Cambridge University Press, 1976.
10. Buckler, J. M. H. *A Reference Manual of Growth and Development*. Oxford, Blackwell Scientific Publications, 1979.
11. Guzmán, M. A., R. Sibrián & R. Flores. Procedimientos básicos en el registro y proceso de datos. *Arch. Latinoamer. Nutr.*, 33(2): 257-267, 1983.
12. Maloff, C. H. & R. W. Zears. *Computers in Nutrition*. Washington, D. C., Ed. Artech, 1979.
13. García-Diz, L., P. Carrasco & I. Goñi. Estudio antropométrico de una población infantil madrileña. *Rev. Clin. Española*, 145: 9-12, 1984.
14. Goñi, I., C. Peña & L. García-Diz. Estudio antropométrico de la composición corporal de una población infantil madrileña. *Rev. Clin. Española*, 1984. (En prensa)
15. Moreiras-Varela, O., A. Carbajal, M. J. Blazquez, L. Cabrera & A. Martínez. La alimentación en la escuela y en el hogar de niños madrileños: estudio piloto. *Rev. Esp. Pediatr.*, 40(4): 257-266, 1984.
16. IUNS. The creation of growth standards, a committee report. *Am. J. Clin. Nutr.*, 25: 218, 1972.
17. FAO/UNICEF/WHO. *Methodology of Nutritional Surveillance*. Geneva, WHO, 1976, p. 20-60. (WHO Technical Report Series No. 53).
18. Sokal, R. R. & F. J. Rohlf. *Biometría*. Madrid, Editorial H. Blume, 1979.
19. Morrison, D. F. *Multivariate Statistical Methods*. 2nd ed. New York, N. Y., McGraw Hill International Book Co., 1978.
20. Ruckdeschel, F. R. *Basic Scientific Subrutines*. Vol I. New York, N. Y., Byte/McGraw Hill Publication Co., 1981.
21. Ruckdeschel, F. R. *Basic Scientific Subrutines*. Vol. II. New York, N. Y., Byte/McGraw Hill Publication Co., 1981.